
Provider is a library which helps managing the operator configuration of a system.
The term operator configuration refers to the settings which have to be provided by the operator of the system before the system is started. Typical examples include database credentials (host, database name, login, password), http and https ports, 3rd party API keys, and such. On the other hand, things such as Ecto database adapter, or dev/test-only variations such as mocks, are not a part of operator configuration.
Provider offers the following features:
	Consolidation of operator settings into a single place
	Basic strong typing (automatic conversion of external settings to string, integer, or boolean)
	Compile time guarantees
	Runtime validation during the system startup
	Automatic creation of operator templates, such as .env file which lists all configuration parameters
	In-code provisioning of dev/test defaults, which removes the need to use external files such as .envrc, or manage duplicate settings in CI yamls

Provider was born out of real need and is based on lessons learned and issues experienced with config scripts and app config. Provider doesn't use app config, and is instead only focused on fetching values from external sources, such as OS env. The responsibility of working with those values is left to the client.
Provider is currently used in a few small projects, but it's not exhaustively tested nor optimized. The API is not considered as stable, and it may change significantly. Provider currently only supports the features needed by the projects it powers. Therefore, it has a couple of limitations:
	Provider can't be used to provide compile-time settings or to configure applications which require the settings before the main app is started (such as Kernel or Logger). In these cases you'll still need to use config.exs & friends.
	Only OS env is currently supported as an external source.
	No support for nil or empty strings. Each configured value has to be provided, and it can't be an empty string.

Tackling these shortcomings is on the roadmap, but it hasn't happen yet, because there was no actual need to address them so far.
Quick start
Add provider as a dependency:
defmodule MySystem.MixProject do
  # ...

  defp deps do
    [
      {:provider, "~> 0.1"},
      # ...
    ]
  end
end
Create a module where you'll consolidate your configuration:
defmodule MySystem.Config do
  use Provider,
    source: Provider.SystemEnv,
    params: [
      # The `:dev` option sets a dev/test default. This default won't be used in `:prod` mix env.
      {:db_host, dev: "localhost"},

      # The `:test` option overrides the default in the test mix env.
      {:db_name, dev: "my_db_dev", test: "my_db_test"},

      # The `:default` option sets a default in all mix envs, which means that this setting is
      # optional.
      #
      # Owing to `type: :integer`, the external input will be converted into an integer. An error
      # will be raised if this conversion fails.
      {:db_pool_size, type: :integer, default: 10},

      # ...
    ]
end
Validate configuration during app startup:
defmodule MySystem.Application do
  use Application

  def start(_type, _args) do
    MySystem.Config.validate!()

    # ...
  end

  # ...
end

Use config module functions to fetch the values:
defmodule MySystem.Repo do
  use Ecto.Repo,
    otp_app: :my_system,
    adapter: Ecto.Adapters.Postgres

  @impl Ecto.Repo
  def init(_type, config) do
    {:ok,
      Keyword.merge(
        config,
        hostname: MySystem.Config.db_host(),
        database_name: MySystem.Config.db_name(),
        pool_size: MySystem.Config.db_pool_size(),
        # ...
      )
    }
  end
end
Invoke MySystem.Config.template() in prod Mix env to generate a .env template with all configuration parameters. The function will return a string which you can print to screen or save to a file.
Alternatives
There are a couple of other libraries available with similar features:
	Skogsrå
	Specify
	Vapor

License

Copyright
  2020-2023, [Very Big Things](https://verybigthings.com/)
  2023-      Saša Jurić

Permission is hereby granted, free of charge, to any person obtaining a copy of this software and associated documentation files (the "Software"), to deal in the Software without restriction, including without limitation the rights to use, copy, modify, merge, publish, distribute, sublicense, and/or sell copies of the Software, and to permit persons to whom the Software is furnished to do so, subject to the following conditions:

The above copyright notice and this permission notice shall be included in all copies or substantial portions of the Software.

THE SOFTWARE IS PROVIDED "AS IS", WITHOUT WARRANTY OF ANY KIND, EXPRESS OR IMPLIED, INCLUDING BUT NOT LIMITED TO THE WARRANTIES OF MERCHANTABILITY, FITNESS FOR A PARTICULAR PURPOSE AND NONINFRINGEMENT. IN NO EVENT SHALL THE AUTHORS OR COPYRIGHT HOLDERS BE LIABLE FOR ANY CLAIM, DAMAGES OR OTHER LIABILITY, WHETHER IN AN ACTION OF CONTRACT, TORT OR OTHERWISE, ARISING FROM, OUT OF OR IN CONNECTION WITH THE SOFTWARE OR THE USE OR OTHER DEALINGS IN THE SOFTWARE.

Retrieval of configuration settings from external sources, such as OS env vars.
This module is an alternative to app env for retrieval of configuration settings. It allows you
to properly consolidate system settings, define per-env defaults, add strong typing, and
compile time guarantees.
Basic example
defmodule MySystem.Config do
  use Provider,
    source: Provider.SystemEnv,
    params: [
      {:db_host, dev: "localhost"},
      {:db_name, dev: "my_db_dev", test: "my_db_test"},
      {:db_pool_size, type: :integer, default: 10},
      # ...
    ]
end
This will generate the following functions in the module:
	fetch_all - retrieves values of all parameters
	validate! - validates that all parameters are correctly provided
	db_host, db_name, db_pool_size, ... - getter of each declared parameter

Describing params
Each param is described in the shape of {param_name, param_spec}, where param_name is an
atom, and param_spec is a keyword list. Providing only param_name (without a tuple), is the
same as {param_name, []}.
The following keys can be used in the param_spec:
	:type - Param type (see type/0). Defaults to :string.
	:default - Default value used if the param is not provided. Defaults to nil (no default).
	:dev - Default value in :dev and :test mix env. Defaults to nil (no default).
	:test - Default value in :test mix env. Defaults to nil (no default).

Default options are considered in the following order:
	:test (if mix env is :test)
	:dev (if mix env is either :dev or :test)
	:default

For example, if :test and :default options are given, the :test value will be used as a
default in :test env, while :default will be used in all other envs.
When you invoke the generated functions, values will be retrieved from the external storage
(e.g. OS env). If some value is not available, a default value will be used (if provided). The
values are then casted according to the parameter type.
Each default can be a constant, but it can also be an expression, which is evaluated at runtime.
For example:
defmodule MySystem.Config do
  use Provider,
    source: Provider.SystemEnv,
    params: [
      # db_name/0 will be invoked when you try to retrieve this parameter (or all parameters)
      {:db_name, dev: db_name()},
      # ...
    ]

  defp db_name(), do: #...
end
It's worth noting that Provider performs compile-time purging of needless defaults. When you
compile the code in :prod, :dev and :test defaults will not be included in the binaries.
Consequently, any private function invoked only in dev/test will also not be invoked, and you'll
get a compiler warning when compiling the code in prod. To eliminate such warnings, you can
conditionally define the function only in required mix env, by moving the function definition
under an if Mix.env() == conditional.
Generating template
The config module will contain the template/0 function which generates the configuration
template. To print the template to stdout, you can invoke:
MIX_ENV=prod mix compile
MIX_ENV=prod mix run --no-start -e 'IO.puts(MySystem.Config.template())'
Lower level API
The basic retrieval functionality is available via functions of this module, such as
fetch_all/2, or fetch_one/2. These functions are a lower level plumbing API which is less
convenient to use, but more flexible. Most of the time the use-based interface will serve you
better, but if you have more specific needs which are not covered by that, you can reach for
these functions.

          @spec fetch_one!(source(), param_name(), param_spec()) :: value()


      


Retrieves a single param, raising if the value is not available.

          @callback display_name(Provider.param_name()) :: String.t()


      


Invoked to convert the param name to storage specific name.

          @callback template(Provider.params()) :: String.t()


      


Invoked to create operator template.

          @callback values([Provider.param_name()]) :: [Provider.value()]


      


Invoked to provide the values for the given parameters.
This function should return all values in the requested orders. For each param which is not
available, nil should be returned.

Provider source which retrieves values from OS env vars.

